Integration by Fourier Transform—INFR

The program INFR (**In**tegration by **F**ou**r**ier Transform) is a subroutine subprogram that calculates the discrete value of an integral $\int_0^{t=m\Delta t} x(t)dt\ (m = 0,1,2,\cdots,N-1)$ by using the Fourier transform and the inverse Fourier transform when the equally spaced time history $x(t)$ is given as N discrete-valued data $x_m(m = 0,1,2,\cdots,N-1)$.

INFR（**In**tegration by **F**ou**r**ier Transform）

【Purpose】

To integrate equally spaced time history data using the Fast Fourier Transform.

【Usage】

(1)　How to connect

```
CALL INFR (N, X, ND, DT)
```

Argument	Type	Parameter in calling program	Return Parameter
N	I	Total number of 1-D array X	Unchanged
X	R 1-D array (ND)	Time history data to be integrated	Integrated time history data
ND	I	Dimension size of X in calling program (ND .GE. N)	Unchanged
DT	R	Time increment (unit: sec)	Unchanged

(2)　Necessary subroutines and function subprograms

FAST

(3)　Remarks

N must be less than or equal to 8192. (But it can be changed easily.)

【Calculation Method】

When N discrete-valued data x_m (m=0, 1, 2, $\cdots$,N-1) at equal interval Δt are given in the time domain, first Fourier transform them to obtain the complex Fourier coefficients C_k in the frequency domain.

$$\left.\begin{aligned} S_0 &= 2\left[\frac{(N-1)C_0}{2}\left(\frac{\pi}{N}\right) - \sum_{k=1}^{N/2-1}\frac{\mathrm{J}(C_k)}{k}\right] \\ S_k &= \left[-1+\mathrm{i}\cot\left(\frac{\pi}{N}\right)k\right]C_0\left(\frac{\pi}{N}\right) - \frac{\mathrm{i}\,C_k}{k}, \quad S_{N-k} = S_k^{*} \quad k = 1, 2, \cdots, N/2-1 \\ S_{N/2} &= -\left(\frac{\pi}{N}\right)C_0 \end{aligned}\right\}$$

Then, after performing the following operations, the discrete value of the integral can be obtained by performing the inverse Fourier transform.

$$\left(\int_0^t x\,\mathrm{d}t\right)_m = \frac{N\Delta t}{2\pi}\sum_{k=0}^{N-1} S_k\,\mathrm{e}^{\mathrm{i}(2\pi k m/N)} \qquad m = 0, 1, 2, \cdots, N-1$$

$$\left(\int_0^t x\,\mathrm{d}t\right)_m = \int_0^{t=m\Delta t} x(t)\,\mathrm{d}t \qquad m = 0, 1, 2, \cdots, N-1$$

The fast Fourier transform program FAST is used for the Fourier transform and the inverse Fourier transform. To make it convenient to use the Fourier, add a trailing zero to make the number of data a power of 2, and then complex the data by the 'INITIALIZATION' block in the program. At the end of the Fourier transform, the complex Fourier coefficients C_k are all multiplied by N, but the correction is done after the inverse Fourier transform is completed.

【Program List】

```
C  * * * * * * * * * * * * * * * * * * * * * * * * * * *
C     SUBROUTINE FOR INTEGRATION BY FOURIER TRANSFORM
C  * * * * * * * * * * * * * * * * * * * * * * * * * * *
C
C                                CODED BY Y.OHSAKI
C
C     PURPOSE
C       TO INTEGRATE AN EQUI-SPACED TIME HISTORY BY APPLICATION OF
C       FAST FOURIER TRANSFORM
C
C     USAGE
C       CALL INFR(N,X,ND,DT)
C
C     DESCRIPTION OF ARGUMENTS
C       N     - TOTAL NUMBER OF DATA   N.LE.8192
C       X(ND) - ORIGINAL/INTEGRATED DATA AT CALL/RETURN
C       ND    - DIMENSION OF X IN CALLING PROGRAM
C       DT    - TIME INCREMENT IN DATA
C
C     SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED
C       FAST
C
      SUBROUTINE INFR(N,X,ND,DT)
C
      COMPLEX   C(8192)
```

```
      DIMENSION X(ND)
      PARAMETER (P2=6.283185)
C
C     INITIALIZATION
C
      DO 110 M=1,N
      C(M)=CMPLX(X(M),0.)
  110 CONTINUE
      NT=2
  120 IF(NT.GE.N) GO TO 130
      NT=NT*2
      GO TO 120
  130 IF(NT.EQ.N) GO TO 150
      DO 140 M=N+1,NT
      C(M)=(0.,0.)
  140 CONTINUE
  150 NFOLD=NT/2+1
      PN=P2/2./REAL(NT)
C
C     FOURIER TRANSFORM
C
      CALL FAST(NT,C,8192,-1)
C
C     INTEGRATION
C
      C1=REAL(C(1))
      S1=REAL(NT-1)/2.*C1*PN
      DO 160 K=2,NFOLD-1
      S1=S1-AIMAG(C(K))/REAL(K-1)
      C(K)=CMPLX(-1.,1./TAN(REAL(K-1)*PN))*C1*PN-(0.,1.)*C(K)/REAL(K-1)
      C(NT-K+2)=CONJG(C(K))
  160 CONTINUE
      C(1)=CMPLX(S1*2.,0.)
      C(NFOLD)=CMPLX(-C1*PN,0.)
C
C     FOURIER INVERSE TRANSFORM
C
      CALL FAST(NT,C,8192,+1)
      DO 170 M=1,N
      X(M)=REAL(C(M))/P2*DT
  170 CONTINUE
      RETURN
      END
```

【Example】

Read the acceleration time history from the file EQ.01, integrate it with the subroutine **INFR** to obtain the velocity time history, and then differentiate it with the subroutine **DIFR** to return to the original acceleration time history.

```
C
      DIMENSION DATA(800),VEL(800),ACC(800),DYY(800)
C
```

```
      READ(5,501) DT,NN,(DATA(M),M=1,NN)
      CALL INFR (NN,DATA,800,DT)
      DO 110 M=1,NN
      VEL(M)=DATA(M)
  110 CONTINUE
      CALL DIFR (NN,DATA,800,DT)
      DO 120 M=1,NN
      ACC(M)=DATA(M)
  120 CONTINUE
      STOP
C
  501 FORMAT(T51,F10.0,I10/(8F10.0))
      END
```

Output: The integrated data is stored in array *VEL*, and the differentiated data is stored in array *ACC*. These are shown in the following figures in the order of the original acceleration time history, the integrated velocity time history, and the differentiated acceleration time history.